


バランス 2 輪車の制御シミュレーション



倒立振り子 2 輪車



タイヤ シャーシ

車輪にたいして

タイヤ質量

$$M_T = 0.17 \times 2 = 0.34 \text{ kg}$$

$$\text{慣性モーメント } J_T = 3.4 \times 10^{-3} \text{ kgm}^2$$

$$\text{外周半径 } R = 0.1 \text{ m}$$

シャーシにたいして

$$\text{下質量 } m_1 = 0.5 \text{ kg}$$

$$\text{上質量 } m_2 = 1 \text{ kg}$$

$$\text{長さ } l = 0.4 \text{ m}$$

$$M = M_T + m_1 + m_2 + J_T / R^2 = 2.18 \text{ kg}$$

$g = 9.8 \text{ m/s}^2$ としてこれらを次の状態方程式の右辺の第 1 項に入れる。

$$\begin{pmatrix} \dot{x} \\ \ddot{x} \\ \dot{\theta} \\ \ddot{\theta} \end{pmatrix} = \begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & -\frac{m_2}{M-m_2}g & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & \frac{M}{M-m_2} \frac{g}{l} & 0 \end{pmatrix} \begin{pmatrix} x \\ \dot{x} \\ \theta \\ \dot{\theta} \end{pmatrix} + \begin{pmatrix} 0 \\ \frac{1}{M-m_2} \left(\frac{1}{R} + \frac{1}{l} \right) \\ 0 \\ -\frac{1}{M-m_2} \frac{1}{l} \left(\frac{1}{R} + \frac{M}{m_2} \frac{1}{l} \right) \end{pmatrix} C$$

ここで状態行列に数値を入れて次を得る。

$$A = \begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & -2.917 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 31.792 & 0 \end{pmatrix} \quad B = \begin{pmatrix} 0 \\ 7.44 \\ 0 \\ -31.101 \end{pmatrix}$$

$K = (k_1 \ k_2 \ k_3 \ k_4) = (0.17, 0.21, 1.64, 0.24)$ とおく。

傾き角の初期値 $= 0.1 \text{ rad}$ を与えて、 $dX = (A + BK)X$ を使いこのシステムを

シミュレートしよう。

```

!balancing scooter
!simul02.bas

set window -1,10,-0.2,0.2
set axis color 1
draw grid(1,0.02)

dim A(4,4),B(4,1),C(4,4),dx(4),x(4)
dim K(1,4),BK(4,4),F(4)

let dt=0.001

let A(1,2)=1
let A(2,3)=-2.917
let A(3,4)=1
let A(4,3)=31.792


let B(2,1)=7.44
let B(4,1)=-31.101

data 0.17,0.21,1.64,0.24
mat read K

set line width 3
let x(3)=0.1
for t=0 to 10 step dt
  mat BK=B*K
  mat C=A+BK
  mat F=C*x
  mat dx=dt*F
  mat x=x+dx
  plot lines:t,x(1)
  plot lines:t,x(3)
next t

END

```



次にデフォルメしたアニメに挑戦しよう。 わかりやすくするために、移動距離は大きさにしている。 今までの計算では、シャーシの傾き角は時計周りを正とした。 描画では、角度は反時計周りを正とするので、傾き角 (x_3) にマイナス記号をつける。

```
!anim02.bas animation

set window -10,10,-8,12
set axis color 1

dim A(4,4),B(4,1),C(4,4),dx(4),x(4)
dim K(1,4),BK(4,4),F(4)


let dt=0.02

let A(1,2)=1
let A(2,3)=-2.917
let A(3,4)=1
let A(4,3)=31.792

let B(2,1)=7.44
let B(4,1)=-31.101

data 0.17,0.21,1.64,0.24
mat read K

for i=0 to 5
  let x(3)=0.1
  for t=0 to 5 step dt
    set draw mode hidden
    clear
    mat BK=B*K
    mat C=A+BK
    mat F=C*x
    mat dx=dt*F
    mat x=x+dx
    draw segway with rotate(-x(3))*shift(x(1)*40,0)
    draw tire with shift(x(1)*40,0)
    plot lines:-10,-3;10,-3
    set draw mode explicit
    wait delay 0.01
  next t
next i
END
```



```
external picture segway
draw disk with scale(1)*shift(0,9)
plot area:-0.3,0;0.3,0;0.3,9;-0.3,9;-0.3,0
draw disk with scale(1.5)
set area color 0
draw disk with scale(0.5)
set area color 1
end picture
external picture tire
set line color 8
draw circle with scale(3)
end picture
```